

A Package on Formal Power Series

WOLFRAM KOEPF

Konrad-Zuse-Zentrum für Informationstechnik
Heilbronner Str. 10
D-10711 Berlin
koepf@zib-berlin.de

Mathematica Journal **4**, to appear in May, 1994

Abstract:

Formal Laurent-Puiseux series are important in many branches of mathematics. This paper presents a *Mathematica* implementation of algorithms developed by the author for converting between certain classes of functions and their equivalent representing series. The package **PowerSeries** handles functions of rational, exponential, and hypergeometric type, and enables the user to reproduce most of the results of Hansen's extensive table of series. Subalgorithms of independent significance generate differential equations satisfied by a given function and recurrence equations satisfied by a given sequence.

Scope of the Algorithms

A common problem in mathematics is to convert an expression involving elementary or special functions into its corresponding formal Laurent-Puiseux series of the form

$$f(x) = \sum_{k=k_0}^{\infty} a_k x^{k/n}. \quad (1)$$

Expressions created from algebraic operations on series, such as addition, multiplication, division, and substitution, can be handled by finite algorithms if one truncates the resulting series. These algorithms are implemented in *Mathematica*'s **Series** command. For example:

```
In[1]:= Series[Sin[x] Exp[x], {x, 0, 5}]
```

$$\text{Out[1]} = x + x^2 + \frac{x^3}{3} + \frac{x^4}{6} + \frac{x^5}{30} + 0[x]$$

It is usually much more difficult to find the exact formal result, that is, an explicit formula for the coefficients a_k .

This paper presents the package **PowerSeries**, a *Mathematica* implementation of algorithms developed by the author ([3], [4], [5]) for computing Laurent-Puiseux series for three types of functions: functions of *rational type*, which are rational or have a rational derivative of some order; functions of *exponential type*, which satisfy a homogeneous linear differential equation with constant coefficients; and functions of *hypergeometric type*, which have a representation (1) with coefficients satisfying a recurrence equation of the form

$$\begin{aligned} a_{k+m} &= R(k) a_k & \text{for } k \geq k_0 \\ a_k &= A_k & \text{for } k = k_0, k_0 + 1, \dots, k_0 + m - 1 \end{aligned} \quad (2)$$

for some rational function R , $A_k \in \mathbb{C}$, $A_{k_0} \neq 0$, and some positive integer m , called the *symmetry number* of (the given representation of) the function.

To find the Laurent-Puiseux expansion of a function $f(x)$ about the point x_0 , one uses the function call **PowerSeries**[$f(x)$, $\{x, x_0\}$]. The package also extends the built-in **Series** command in the case that its second argument is a list of length two:

```
In[2]:= << PowerSeries.m
ps-info: PowerSeries version 1.02, Mar 07, 1994

In[3]:= Series[Sin[x] Exp[x], {x, 0}]

      k/2  k      k Pi
      2    x Sin[----]
                4
Out[3]= Sum[-----, {k, 0, Infinity}]
          k!
```

Our algorithms also handle the converse problem of finding a “closed form” representation of a given Laurent-Puiseux series (that is, the generating function of the sequence a_k). The standard *Mathematica* packages **Algebra**’**SymbolicSum**’ and **DiscreteMath**’**RSolve**’ also deal with this problem, using different approaches. We will discuss the connection of our package with these packages and other *Mathematica* functions at the end of this introduction.

The most interesting case is formed by the functions of hypergeometric type, which include almost all transcendental elementary functions like x^n , **Exp**, **Log**, and the trigonometric and inverse trigonometric functions; all kinds of special functions like the Airy functions, the Bessel functions, the integral functions **Erf**, **ExpIntegralEi**, **CosIntegral**, and **SinIntegral**, the associated Legendre functions **LegendreP** and **LegendreQ**, the hypergeometric functions **HypergeometricU**, **Hypergeometric0F1**, **Hypergeometric1F1**, as well as **Hypergeometric2F1**, the orthogonal polynomials **JacobiP**, **GegenbauerC**, **ChebyshevT**, **ChebyshevU**, **LegendreP**, **LaguerreL**, and **HermiteH**; and many more functions.

Our algorithms also handle the following special functions. Since they are not built in to *Mathematica*, we have included the necessary definitions in our package: the Bateman functions `Bateman[n,x]` ([1], (13.6), and [8]); the Hankel functions `Hankel1[n,x]` and `Hankel2[n,x]` ([1], (9.1)); the Kummer and Whittaker functions `KummerM[a,b,x]`, `KummerU[a,b,x]`, `WhittakerM[a,b,x]`, `WhittakerW[a,b,x]` ([1], (13.1)); the Struve functions `StruveH[n,x]` and `StruveL[n,x]` ([1], Chapter 12); the hypergeometric functions `Hypergeometric1F0[a,x]`, `Hypergeometric2F0[a,b,x]`; the iterated integrals of the complementary error function `Erfc[n,x]` ([1], (7.2)); the Abramowitz functions `Abramowitz[n,x]` ([1], (27.5)); and the repeated integrals of the normal probability integral `NormalIntegral[n,x]` ([1], (26.2.41)). Informations about these functions can be obtained by using *Mathematica*'s `?` feature after the package is loaded.

Our algorithms depend strongly on the fact that any function of hypergeometric type satisfies a *simple differential equation*, i. e., a homogeneous linear differential equation with polynomial coefficients (see [3], Theorems 2.1, and 8.1). One of our main algorithms finds this differential equation for (practically) any expression for which such an equation exists.

On the other hand, it is similarly important that almost every function that we may write down satisfies a simple differential equation. All functions that can be constructed algebraically from the functions mentioned above by addition, multiplication, and composition with rational functions and rational powers, satisfy such a differential equation ([10], [9]).

The conversion of functions to and from their representing Laurent-Puiseux expansions is described in detail in ([3], [5]). Here, we give a brief description and some examples for each of the types of functions considered.

To find the Laurent-Puiseux expansion of an expression $f(x)$, the function `PowerSeries` performs the following steps. First, a homogeneous linear differential equation with polynomial coefficients for f is generated. (If no low-order differential equation exists, this may take some time.) This differential equation is converted to an equivalent recurrence equation for a_k , an easy and fast procedure. If the recurrence equation is of the hypergeometric type, the coefficients can be calculated explicitly from a finite number of initial values. (Finding the initial values may require the calculation of limits. This procedure may be slow, or may fail.)

As an example of a function of rational type, let's take $f(x) = \arctan x$. Here, $f'(x) = \frac{1}{1+x^2}$ is rational, so we apply a special algorithm for the rational case, and integrate. First, the complex partial fraction decomposition is derived:

$$f'(x) = \frac{1}{1+x^2} = \frac{1}{2} \left(\frac{1}{1+ix} + \frac{1}{1-ix} \right) ,$$

from which the coefficients b_k of the derivative $F'(x) = \sum_{k=0}^{\infty} b_k x^k$ can be deduced:

$$b_k = \frac{1}{2} \left(i^k + (-i)^k \right) = \frac{i^k}{2} \left(1 + (-1)^k \right) ,$$

Finally, after an integration, we get

$$f(x) = \sum_{k=0}^{\infty} \frac{i^k (1 + (-1)^k)}{2 (k+1)} x^{k+1} .$$

To illustrate the hypergeometric case, we again take $f(x) = \arctan x$. We have $f'(x) = \frac{1}{1+x^2}$ and $f''(x) = -\frac{2x}{(1+x^2)^2}$, so we arrive at the differential equation

$$(1+x^2) f''(x) + 2x f'(x) = 0 .$$

This differential equation can be converted to the recurrence equation

$$(k+1)(k+2) a_{k+2} + k(k+1) a_k = 0 \quad (3)$$

by the rule-based function `detore[DE,F,x,a,k]`, which takes as input the left hand side of the differential equation for the function $F[x]$ and computes the left hand side of the recurrence equation for the coefficients $a[k]$:

```
detore[g_ + h_, F_, x_, a_, k_] :=
  detore[g, F, x, a, k] + detore[h, F, x, a, k]
detore[c_ * g_, F_, x_, a_, k_] := c * detore[g, F, x, a, k] /;
  FreeQ[c, x] && FreeQ[c, F]
detore[Derivative[k0_][F_][x_], F_, x_, a_, k_] :=
  Pochhammer[k+1, k0] * a[k+k0]
detore[F_[x_], F_, x_, a_, k_] := a[k]
detore[x_^j_. * Derivative[k0_][F_][x_], F_, x_, a_, k_] :=
  Pochhammer[k+1-j, k0] * a[k+k0-j]
detore[x_^j_. * F_[x_], F_, x_, a_, k_] := a[k-j]

In[4] := Simplify[detore[2*x*f'[x] + f''[x] + x^2*f''[x], f, x, a, k]]

Out[4] = (1 + k) (k a[k] + 2 a[2 + k] + k a[2 + k])
```

As a last step, this recurrence equation is solved, resulting in the representation

$$f(x) = \sum_{k=0}^{\infty} \frac{(-1)^k}{2k+1} x^{2k+1} .$$

Note that in most cases the hypergeometric-type result, if applicable, is the simplest one.

Finally, let us take $f(x) = \sin x e^x$, the function whose series was calculated above. Then $f'(x) = e^x (\sin x + \cos x)$ and $f''(x) = 2e^x \cos x$. If we look for an equation of the form

$$f'' + A_1 f' + A_0 f = 2e^x \cos x + A_1 e^x (\sin x + \cos x) + A_0 e^x \sin x = 0 ,$$

we may expand and recombine rationally dependent expressions to get

$$f'' + A_1 f' + A_0 f = (2 + A_1) e^x \cos x + (A_1 + A_0) e^x \sin x = 0 .$$

This equation must be valid for all x , so the coefficients of $e^x \cos x$ and $e^x \sin x$ must vanish. Hence $A_1 = -2$, $A_0 = 2$, and we have the differential equation for f :

$$f'' - 2f' + 2f = 0 .$$

This differential equation has constant coefficients, so f is of exponential type. For $b_n = n! a_n$, we have the recurrence equation

$$b_{n+2} - 2b_{n+1} + 2b_n = 0$$

with initial values

$$b_0 = a_0 = f(0) = 0 \quad \text{and} \quad b_1 = a_1 = \frac{f'(0)}{1!} = 1 ,$$

which generates the solution

$$\begin{aligned} a_n &= \frac{b_n}{n!} = \frac{1}{n!} \frac{(1+i)^n - (1-i)^n}{2i} \\ &= \frac{1}{n!} \operatorname{Im} (1+i)^n = \frac{1}{n!} \operatorname{Im} \left(\sqrt{2} e^{i\frac{\pi}{4}} \right)^n \\ &= \frac{1}{n!} 2^{\frac{n}{2}} \sin \frac{n\pi}{4} , \end{aligned}$$

Hence

$$f(x) = \sum_{n=0}^{\infty} \frac{1}{n!} 2^{\frac{n}{2}} \sin \frac{n\pi}{4} x^n .$$

We will present more examples to demonstrate our *Mathematica* implementation later in the paper. Additional examples are also given in [4].

To find the generating function of a given sequence, the process can be reversed. The function **Convert** provides the following procedure to calculate the closed-form representation of $\sum_{k=k_0}^{\infty} a_k x^{k/n}$. First, a homogeneous linear recurrence equation with polynomial coefficients for a_k is generated. (If no low order recurrence equation exists, this may take some time.) This recurrence equation is converted to an equivalent differential equation for the generating function, also an easy and fast procedure. Finally, the differential equation is solved using appropriate initial values. (This step depends heavily on *Mathematica*'s built-in differential equation solver **DSolve**. It often takes a lot of time to solve the differential equation, or may also fail. Moreover, the calculation of limits is needed to find the initial values. This part of the procedure may be slow or fail, as well.)

As an example, we try to find the generating function

$$\sum_{k=0}^{\infty} \frac{(-1)^k}{2k+1} x^k$$

of the sequence $a_k = \frac{(-1)^k}{2k+1}$. Taking a shift, we get the relation $a_{k+1} = \frac{(-1)^{k+1}}{2k+3}$ so that the recurrence equation

$$(2k+3)a_{k+1} + (2k+1)a_k = 0$$

is established for $k \geq 0$. If we multiply this equation by the factor $(k+1)$, the resulting recurrence equation

$$(k+1)(2k+3)a_{k+1} + (k+1)(2k+1)a_k = 0 \quad (4)$$

holds for all integers k . The back-substitution can be done by the *Mathematica* procedure `retode[re,n,f,x]`, defined by these rules:

```
theta[f_, x_] := x D[f, x]
retode[eq1_ + eq2_, k_, f_, x_] :=
  retode[eq1, k, f, x] + retode[eq2, k, f, x]
retode[c_ * eq_, k_, f_, x_] := c * retode[eq, k, f, x] /;
  (FreeQ[c, k] && FreeQ[c, f] && FreeQ[c, x])
retode[a[k_ + m_.], k_, f_, x_] := f[x]/x^m
retode[k^j_. * a[k_ + m_.], k_, f_, x_] :=
  theta[retode[k^(j-1) * a[k+m], k, f, x], x]
retode[p_ * a[k_ + m_.], k_, f_, x_] :=
  retode[Expand[p * a[k+m]], k, f, x] /; PolynomialQ[p]
```

Applied to the left hand side of (4), `retode` gives:

```
In[5] := Simplify[retode[(k+1)(2k+3)a[k+1] + (k+1)(2k+1)a[k], k, f, x]]
```

```
Out[5] = f[x] + 3 f'[x] + 5 x f'[x] + 2 x f''[x] +
> 2 x^2 f''[x]
```

The initial value problem

$$(2x + 2x^2)f'' + (3 + 5x)f' + f = 0 ,$$

$$f(0) = a_0 = 1 , \quad f'(0) = a_1 = -\frac{1}{3} ,$$

has the solution

$$f(x) = \frac{\arctan \sqrt{x}}{\sqrt{x}} .$$

Note that *Mathematica* Version 2.0 gives the correct result:

```
In[6] := Convert[Sum[(-1)^k/(2k+1)*x^k, {k, 0, Infinity}], x]
```

```
Out[6] = -----
          ArcTan[Sqrt[x]]
          Sqrt[x]
```

whereas the `DSolve` command of *Mathematica* Version 2.2 gives a wrong result.

Now we want to give some remarks on the connection of our package with several standard *Mathematica* packages.

In several instances, `PowerSeries` depends on the calculation of limits, so it may be helpful to load the package `Calculus`Limit``. Since the `Convert` function of `PowerSeries` depends heavily on the `DSolve` command, it may also be helpful to load `Calculus`DSolve`` (Version 2.2). In this case, the incorrect result mentioned above does not occur.

The aim of the package `Algebra`SymbolicSum`` is to give closed form representations of sums, in particular, sums of the generating function type. Thus, it is not surprising that many of the conversions that can be obtained by our `Convert` function are also obtained by the `SymbolicSum` package. Among the expressions that cannot be handled by the `SymbolicSum` package are the sums `conv[4]` and `conv[5]` defined below, and, in general, sums involving special functions. Typically, our implementation is slower as it always uses `DSolve`, which takes a lot of time. The advantage is that the procedure forms an algorithmic treatment of the problem, and is not based on heuristics.

The purpose of the package `DiscreteMath`RSolve`` is to solve recurrence equations. It uses generating function techniques and defines the function `GeneratingFunction` to find the generating function of a recurrence equation. The procedure uses heuristic techniques, and in particular treats hypergeometric recurrence equations (equation (2) with $m = 1$), but the general case ($m > 1$) is not implemented. Similar restrictions apply to the `RSolve` command.

We give an example: Our function `SimpleRE` generates the recurrence equation (3) for the coefficients of the series representation of a function:

```
In[7]:= re=SimpleRE[ArcTan[x], x]
```

```
Out[7]= k (1 + k) a[k] + (1 + k) (2 + k) a[2 + k] == 0
```

We load the package and try to solve the recurrence equation:

```
In[8]:= << DiscreteMath`RSolve`
```

```
In[9]:= RSolve[{re, a[0]==0, a[1]==1}, a[k], k]
```

```
DSolve::dnim:
```

```
Built-in procedures cannot solve this differential
equation.
```

```
Out[9]= RSolve[{re, a[0] == 0, a[1] == 1}, a[k], k]
```

(This is the *Mathematica* Version 2.2 result. In Version 2.0, another error message occurs.) We see that `RSolve` tries to solve the differential equation for the generating function of the sequence a_k given by the recurrence equation

$$k(1+k)a_k + (1+k)(2+k)a_{k+2} = 0. \quad (5)$$

This problem is too hard. In our treatment, the recurrence equation is replaced by the simpler equation

$$k(1+k)a_k + (1+k)(2+k)a_{k+1} = 0$$

whose generating function is easier to find. Further, our function `PowerSeries` solves the recurrence equation (5) by a direct application of the hypergeometric theory, and splits the series into two parts (odd and even), one of which disappears as a result of a vanishing initial value:

```
In[10]:= Series[ArcTan[x], {x, 0}]
```

```

      k  1 + 2 k
    (-1)  x
Out[10]= Sum[-----, {k, 0, Infinity}]
      1 + 2 k

```

Generation of Differential Equations

In this section, we present some examples of the function `SimpleDE`, which generates the homogeneous linear differential equation of least order satisfied by a given function.

```
In[11]:= de[1] = SimpleDE[x^n * E^(alpha x), x]
```

```
Out[11]= (-n - alpha x) F[x] + x F'[x] == 0
```

```
In[12]:= de[2] = SimpleDE[((1+x)/(1-x))^n, x]
```

```
Out[12]= 2 n F[x] + (-1 + x) (1 + x) F'[x] == 0
```

```
In[13]:= de[3] = SimpleDE[ArcSin[x^5], x]
```

```

      10      11
Out[13]= (4 + x ) F'[x] + (-x + x ) F''[x] == 0

```

```
In[14]:= de[4] = SimpleDE[ArcSin[x], x]
```

```

      2
Out[14]= x F'[x] + (-1 + x ) F''[x] == 0

```

As noted above, we can create new functions satisfying this kind of differential equation by addition, multiplication, and by the composition with rational functions and rational powers:

```
In[15]:= de[5] = SimpleDE[ArcSin[x]^3, x]
```



```

Out[15]= x F'[x] + (-4 + 7 x ) F''[x] +
      2      (3)      2 2      (4)
> 6 x (-1 + x ) F [x] + (-1 + x ) F [x] == 0

In[16]:= de[6] = SimpleDE[Sin[x]^5, x, 6]

Out[16]= 225 F[x] + 259 F''[x] + 35 F (4)
      [x] +
      (6)
> F [x] == 0

In[17]:= de[7] = SimpleDE[Exp[alpha x] * Sin[beta x], x]

Out[17]= (alpha 2 + beta 2) F[x] - 2 alpha F'[x] +
      2
> F''[x] == 0

```

By default, `SimpleDE` searches for a simple differential equation of order up to five. This setting can be changed by entering the order as a third argument. In case one knows a priori that a differential equation of the given type exists, one may choose the order `Infinity`. Note that there is no simple differential equation of order less than five for $\sin^5 x$, so `SimpleDE[Sin[x]^5, x]` fails.

Our implementation handles orthogonal polynomials and special functions using the general method described in [7]. We can derive the usual differential equations:

```

In[18]:= de[8] = SimpleDE[LaguerreL[n, x], x]

Out[18]= n F[x] + (1 - x) F'[x] + x F''[x] == 0

In[19]:= de[9] = SimpleDE[ChebyshevT[n, x], x]

Out[19]= -(n 2 F[x]) + x F'[x] + (-1 + x ) F''[x] == 0

In[20]:= de[10] = SimpleDE[BesselY[n, x], x]

Out[20]= (-n 2 + x 2) F[x] + x F'[x] + x 2 F''[x] == 0

In[21]:= de[11] = SimpleDE[AiryAi[x], x]

Out[21]= -(x F[x]) + F''[x] == 0

```

By the statement above, we know that much more complicated functions satisfy differential equations:

```
In[22]:= de[12] = SimpleDE[Exp[alpha x] * BesselI[n, x], x]
```

```
Out[22]= (-n2 - alpha x - x2 + alpha2 x2) F[x] +
```

```
> (x - 2 alpha x2) F'[x] + x2 F''[x] == 0
```

```
In[23]:= de[13] = SimpleDE[Sin[m x] * BesselJ[n, x], x]
```

```
Out[23]= (n2 - 5 n4 + 4 n6 - x2 - 2 m2 x2 + 22 n2 x2 -
```

```
> 10 m2 n2 x2 - 12 n4 x2 + 12 m2 n4 x2 - x4 +
```

```
> 6 m2 x4 - 5 m4 x4 + 12 n2 x4 - 24 m2 n2 x4 +
```

```
> 12 m4 n2 x4 - 4 x6 + 12 m2 x6 - 12 m4 x6 +
```

```
> 4 m6 x6) F[x] +
```

```
> (-8 x3 - 4 m2 x3 + 8 n2 x3 + 40 m2 n2 x3 -
```

```
> 8 x5 + 8 m4 x5) F'[x] +
```

```
> (-2 x2 + 10 n2 x2 - 8 n4 x2 + 2 x4 - 6 m2 x4 +
```

```
> 16 n2 x4 - 8 x6 + 8 m4 x6) F''[x] +
```

```
> (-4 x3 + 16 n2 x3 - 8 x5 + 8 m2 x5) F(3)[x] +
```

```
> x4 (-1 + 4 n2 - 4 x2 + 4 m2 x2) F(4)[x] == 0
```

```
In[24]:= de[14] = SimpleDE[LegendreP[n, x]^2, x]
```

```
Out[24]= (-4 n x - 4 n2 x) F[x] +
```

```
> (-2 + 4 n2 + 4 n2 x + 6 x2 - 4 n x2 - 4 n2 x2)
```

```
>      F' [x] + 6 x (-1 + x ) F'' [x] +
```

```
>      (-1 + x ) F' [x] == 0
```

(The calculation of these last two results is very time consuming!)

The functions

$$F_n^{(\alpha)}(x) := e^{-x} L_n^{(\alpha)}(2x) ,$$

where $L_n^{(\alpha)}$ denotes the generalized Laguerre polynomial, were studied by the author in [8]. For these functions, we immediately deduce the differential equation:

```
In[25]:= de[15] = SimpleDE[E^(-x) * LaguerreL[n, alpha, 2x], x]
```

```
Out[25]= (1 + alpha + 2 n - x) F[x] +
```

```
>      (1 + alpha) F' [x] + x F'' [x] == 0
```

We may observe that the coefficient of F' in the differential equation vanishes only in the case $\alpha = -1$, which is studied in [8].

Conversion of Simple Differential Equations to Recurrence Equations

The function `DEtoRE` converts a simple differential equation into a recurrence equation for the coefficients of a corresponding Laurent-Puiseux expansion. The output is given in terms of the symbols a and k representing the coefficient sequence a_k . As examples, we convert some of the differential equations that were derived above.

The example

```
In[26]:= re[1] = DEtoRE[de[2], F, x]
```

```
Out[26]= k a[k] + 2 n a[1 + k] - (2 + k) a[2 + k] == 0
```

shows that the function $\left(\frac{1+x}{1-x}\right)^n$ is not of hypergeometric type, whereas the example

```
In[27]:= re[2] = DEtoRE[de[3], F, x]
```

```
Out[27]= k a[k] - (5 + k) (10 + k) a[10 + k] == 0
```

shows that the function $\arcsin(x^5)$ is of hypergeometric type with symmetry number $m = 10$. Similarly, the Airy function is of hypergeometric type with symmetry number $m = 3$:

```
In[28]:= re[3] = DEtoRE[de[11], F, x]
```

```
Out[28]= -a[k] + (2 + k) (3 + k) a[3 + k] == 0
```

Among the functions of the form $e^{\alpha x} I_n(x)$, where $\alpha \in \mathbb{R}$ and I is the modified Bessel function of the first kind, only the cases $\alpha = 0$ and $\alpha = \pm 1$ represent functions of hypergeometric type, as is seen from the recurrence equation

```
In[29]:= re[4] = DEtoRE[de[12], F, x]
```

```
Out[29]= (-1 + alpha) (1 + alpha) a[k] -
```

```
> alpha (3 + 2 k) a[1 + k] +
```

```
> (2 + k - n) (2 + k + n) a[2 + k] == 0
```

Similarly, the functions of the form $\sin(mx) J_m(x)$, where $(m \in \mathbb{R})$ and J is the Bessel function of the first kind, are of hypergeometric type only for the cases $m = 0$ and $m = \pm 1$, as is seen from the recurrence equation

```
In[30]:= re[5] = DEtoRE[de[13], F, x]
```

```
Out[30]= 4 (-1 + m) (1 + m) a[k] +
```

```
> (-1 + m) (1 + m)
```

```
> (33 + 32 k + 8 k2 + 27 m2 + 32 k m + 8 k2 m2 -
```

```
> 12 n2 + 12 m2 n) a[2 + k] +
```

```
> (-297 - 402 k - 210 k2 - 48 k3 - 4 k4 + 198 m2 +
```

```
> 362 k m2 + 206 k2 m2 + 48 k3 m2 + 4 k4 m2 +
```

```
> 246 n2 + 120 k n2 + 16 k2 n2 + 150 m2 n2 +
```

```
> 40 k m2 n2 - 12 n4 + 12 m2 n4) a[4 + k] +
```

```
> (5 + k - n) (6 + k - n) (5 + k + n) (6 + k + n)
```

```
> (-1 + 2 n) (1 + 2 n) a[6 + k] == 0
```

The function `SimpleRE` combines the two steps `SimpleDE` and `DEtoRE` to calculate the recurrence equation of a Laurent-Puiseux series representation of a function.

Calculation of Laurent-Puiseux Series

In this section, we present some of the results of the procedure `PowerSeries`. Obviously, the program can handle the standard power series representations of the elementary functions:

```
In[31]:= ps[1] = PowerSeries[E^x, x]
```

$$\text{Out[31]} = \text{Sum}\left[\frac{x^k}{k!}, \{k, 0, \text{Infinity}\}\right]$$

```
In[32]:= ps[2] = PowerSeries[Sin[x], x]
```

$$\text{Out[32]} = \text{Sum}\left[\frac{(-1)^k x^{1+2k}}{(1+2k)!}, \{k, 0, \text{Infinity}\}\right]$$

```
In[33]:= ps[3] = PowerSeries[ArcSin[x], x]
```

$$\text{Out[33]} = \text{Sum}\left[\frac{(-1)^k x^{1+2k}}{4^{k+1} (2k)!}, \{k, 0, \text{Infinity}\}\right]$$

Here is a series expanded about the point $x = 1$:

```
In[34]:= ps[4] = PowerSeries[Log[x], {x, 1}]
```

$$\text{Out[34]} = \text{Sum}\left[\frac{(-1)^k (-1+x)^{1+k}}{1+k}, \{k, 0, \text{Infinity}\}\right]$$

It may come as a surprise that the following functions are of hypergeometric type:

```
In[35]:= ps[5] = PowerSeries[Exp[ArcSin[x]], x]
```

$$\frac{x^{2k}}{5^{k+1} 2^{k+1}}$$

```

      4 x      Product[- - 2 jj + jj , {jj, k}]
      4
Out[35]= Sum[-----,
              (2 k)!

> {k, 0, Infinity}] +

      k 1 + 2 k      1      2
      4 x      Product[- - jj + jj , {jj, k}]
      2
> Sum[-----,
      (1 + 2 k)!

> {k, 0, Infinity}]

```

```
In[36]:= ps[6] = PowerSeries[Exp[ArcSinh[x]], x]
```

```

      1 k 2 k
      (-(-)) x (2 k)!
      4
Out[36]= x + Sum[-----, {k, 0, Infinity}]
      2
      (1 - 2 k) k!

```

If one is interested in the intermediate calculations, one may give the input `psprint`, which sets the global variable `PSPrintMessages` to `True`:

```
In[37]:= psprint
```

```

In[38]:= ps[7] = PowerSeries[E^x - 2 E^(-x/2) Cos[Sqrt[3]x/2 - Pi/3], x]
ps-info: PowerSeries version 1.02, Mar 07, 1994
ps-info: 3 step(s) for DE:
      (3)
      -F[x] + F [x] == 0
ps-info: RE for all k >= 0:
      a[3 + k] = a[k]/((1 + k)*(2 + k)*(3 + k))
ps-info: function of hypergeometric type
ps-info: a[0] = 0
ps-info: a[1] = 0
      3
ps-info: a[2] = -
      2
ps-info: a[3] = 0
ps-info: a[4] = 0

```

```

      2 + 3 k
      9 (1 + k) x
Out[38]= Sum[-----, {k, 0, Infinity}]
      (3 + 3 k)!

```

Note that the differential equation in this example is extremely simple.

We now suppress intermediate results using the command `nopsprint`, which sets `PSPrintMessages` to `False`.

```
In[39]:= nopsprint
```

Here is a well-known closed formula for the generating function $f(x) = \sum_{k=0}^{\infty} a_k x^k$ of the Fibonacci numbers a_n defined by the recurrence

$$a_{n+1} = a_n + a_{n-1}, \quad a_0 = 0, \quad a_1 = 1.$$

```
In[40]:= ps[8] = PowerSeries[x/(1 - x - x^2), x]
```

$$\text{Out[40]} = \text{Sum}\left[\frac{\left(2 - \frac{1}{-1 + \text{Sqrt}[5]} - \frac{1}{1 + \text{Sqrt}[5]}\right) x^k}{\text{Sqrt}[5]}, \{k, 0, \text{Infinity}\}\right]$$

```
> {k, 0, Infinity}]
```

Moreover, we can use `SimpleRE` to generate the Fibonacci recursion:

```
In[41]:= re[6] = SimpleRE[x/(1 - x - x^2), x, a, k]
```

$$\text{Out[41]} = (1 + k) a[k] + (1 + k) a[1 + k] -$$

$$(1 + k) a[2 + k] == 0$$

Note that the common factor $(1 + k)$ ensures that the recurrence equation holds for all integers k .

Our implementation covers functions that correspond to hypergeometric type Laurent-Puiseux series, rather than just power series. For examples, see [4].

Here, we want to return to the case of special functions. In the previous section, we discovered that the functions $e^x I_n(x)$ and $\sin(x) J_n(x)$ are of hypergeometric type. For $n = 0$ and 1, we get the following series representations:

```
In[42]:= ps[9] = PowerSeries[E^x * BesselI[0, x], x]
```

$$\text{Out[42]} = \text{Sum}\left[\frac{(-1)^k x^k (2k)!}{2^{k+1} k!}, \{k, 0, \text{Infinity}\}\right]$$

```
In[43]:= ps[10] = PowerSeries[E^x * BesselI[1, x], x]
```

$$\text{Out[43]} = \text{Sum}\left[\frac{(-1)^k x^{1+k} (1+2k)!}{2^{k+1} k! (2+k)!}, \{k, 0, \text{Infinity}\}\right]$$

```
In[44]:= ps[11] = PowerSeries[Sin[x] * BesselJ[0, x], x]
```

$$\text{Out[44]} = \text{Sum}\left[\frac{(-1)^k x^{1+2k} (1+4k)!}{4 (2k)! (1+2k)!}, \{k, 0, \text{Infinity}\}\right]$$

```
> {k, 0, Infinity}]
```

```
In[45]:= ps[12] = PowerSeries[Sin[x] * BesselJ[1, x], x]
```

$$\text{Out[45]} = \text{Sum}\left[\frac{(-1)^k x^{2+2k} (3+4k)!}{4 (1+2k)! (3+2k)!}, \{k, 0, \text{Infinity}\}\right]$$

```
> {k, 0, Infinity}]
```

Many more functions constructed from the given special functions are of hypergeometric type (see, for example, Hansen's extensive table of series [2]), and their Laurent-Puiseux expansions can therefore be found by `PowerSeries`.

Calculation of Generating Functions

The function `Convert` calculates generating functions of series:

```
In[46]:= conv[1] = Convert[Sum[(2k)!/k!^2 x^k, {k, 0, Infinity}], x]
```

$$\text{Out[46]} = \frac{1}{\text{Sqrt}[1 - 4 x]}$$

```
In[47]:= conv[2] = Convert[Sum[k!^2/(2k)! x^k, {k, 0, Infinity}], x]
```

$$\text{Out[47]} = \frac{4}{4 - x} + \frac{\text{Sqrt}[x] \text{ArcSin}\left[\frac{\text{Sqrt}[x]}{2}\right]}{(4 - x)^{3/2}}$$

Again, with `psprint` we get information about the intermediate steps:

```
In[48]:= psprint
```

```
In[49]:= conv[3] = Convert[Sum[k!/(2k)! x^k, {k, 0, Infinity}], x]
```



```

ps-info: PowerSeries version 1.02, Mar 07, 1994
ps-info: 1 step(s) for RE.
      a[-1 + k] + 2*(1 - 2*k)*a[k] == 0
ps-info: DE:
      F[x] + (-2 + x) F'[x] + (-4*x) F''[x] == 0
ps-info: Trying to solve DE ...
ps-info: DSolve't encountered
ps-info: DSolve computes
      C[3] + E^(x/4)*x^(1/2)*

>      (C[1] + (C[2]*(-2 -
>
      E^(x/4)*Pi^(1/2)*x^(1/2)*Erf[x^(1/2)/2]))/
>
      (E^(x/4)*x^(1/2)))
ps-info: expression rearranged:
      E^(x/4)*x^(1/2)*C[1] + C[3] +
>
      E^(x/4)*x^(1/2)*C[2]*Erf[x^(1/2)/2]
ps-info: Calculation of initial values...
ps-info: C[3] == 1
ps-info: C[1] == 0
ps-info: C[2]/Pi^(1/2) == 1/2

      x/4
      E
      Sqrt[Pi] Sqrt[x] Erf[-----]
                               2
Out[49]= 1 + -----
                2

```

Here are two generating function examples involving special functions. (Note that `Convert` can only be successful if the solution is representable by an elementary function; otherwise, *Mathematica* will fail to solve the differential equation produced).

```

In[50]:= conv[4] = Convert[Sum[ChebyshevT[k,x] z^k, {k, 0, Infinity}], z]

ps-info: PowerSeries version 1.02, Mar 07, 1994
ps-info: 2 step(s) for RE.
      a[-2 + k] - 2*x*a[-1 + k] + a[k] == 0
ps-info: DE:
      2 F[z] + 4*(-x + z) F'[z] + (1 - 2*x*z + z^2)\
>
      F''[z] == 0
ps-info: Trying to solve DE ...
ps-info: DSolve computes
      (z*C[1] - C[2] + (z*C[2])/C[1])/
>
      (-1 + 2*x*z - z^2)
ps-info: expression rearranged:
      (z*C[1] + C[2])/(-1 + 2*x*z - z^2)

```

```

ps-info: Calculation of initial values...
ps-info: -C[2] == 1
ps-info: (-2*C[1] - 4*x*C[2])/2 == x

Out[50]= 
$$-\frac{1}{-1 + 2 x z - z^2} + \frac{x z}{-1 + 2 x z - z^2}$$


In[51]:= conv[5] = Convert[Sum[LaguerreL[k,a,x] z^k, {k, 0, Infinity}], z]

ps-info: PowerSeries version 1.02, Mar 07, 1994
ps-info: 2 step(s) for RE.
(-1 + a + k)*a[-2 + k] +

> (1 - a - 2*k + x)*a[-1 + k] + k*a[k] == 0
ps-info: DE:
(-1 - a + x + z + a*z) F[z] + ((-1 + z)^2)\

> F'[z] == 0
ps-info: Trying to solve DE ...
ps-info: DSolve computes
E^(x/(-1 + z))*(-1 + z)^(-1 - a)*C[1]
ps-info: expression rearranged:
E^(x/(-1 + z))*(1 - z)^(-1 - a)*C[1]
ps-info: Calculation of initial values...
ps-info: C[1]/E^x == 1

Out[51]= 
$$E \frac{(x z)/(-1 + z)^{-1 - a}}{(1 - z)}$$


```

Calculation of Recurrence Equations

As described above, the function `Convert` works by generating a recurrence equation for the given coefficients, converting it to a differential equation, and then solving. The procedure for generating the recurrence equation is implemented in the function `FindRecursion`. We give some examples:

```

In[52]:= re[7] = FindRecursion[(1 + (-1)^n)/n, n]

Out[52]= (2 - n) a[-2 + n] + n a[n] == 0

In[53]:= re[8] = FindRecursion[n + (-1)^n, n]

Out[53]= (1 - 2 n) a[-2 + n] - 2 a[-1 + n] +

> (-3 + 2 n) a[n] == 0

In[54]:= re[9] = FindRecursion[(n + (-1)^n)/n^2, n]

```

```

Out[54]= (4 - 12 n + 9 n2 - 2 n3) a[-2 + n] +
>      (-2 + 4 n - 2 n2) a[-1 + n] + (-3 n2 + 2 n3) a[n]
>      == 0

In[55]:= re[10] = FindRecursion[1/(2n+1)!, n]

Out[55]= a[-1 + n] - 2 n (1 + 2 n) a[n] == 0

In[56]:= ps[8][[1]]

Out[56]= 
$$\frac{(2^k \left(\frac{1}{-1 + \text{Sqrt}[5]}\right)^k - \left(\frac{-2}{1 + \text{Sqrt}[5]}\right)^k) x}{\text{Sqrt}[5]}$$


In[57]:= re[11] = FindRecursion[%, k]

Out[57]= -(x2 a[-2 + k]) - x a[-1 + k] + a[k] == 0

In[58]:= re[12] = FindRecursion[E^(-x) LaguerreL[n, alpha, 2x], n]

Out[58]= (-1 + alpha + n) a[-2 + n] +
>      (1 - alpha - 2 n + 2 x) a[-1 + n] + n a[n] == 0

In[59]:= re[13] = FindRecursion[n*LaguerreL[n, alpha, 2x], n]

Out[59]= (1 - alpha - 2 n + alpha n + n2) a[-2 + n] +
>      (-2 + 2 alpha + 5 n - alpha n - 2 n2 - 4 x +
>      2 n x) a[-1 + n] + (2 - 3 n + n2) a[n] == 0

In[60]:= re[14] = FindRecursion[LaguerreL[n, alpha, 2x]/n, n]

Out[60]= (2 - 2 alpha - 3 n + alpha n + n2)
>      a[-2 + n] + (-1 + alpha + 3 n - alpha n -
>      2
2

```

```

>      2 n  - 2 x + 2 n x) a[-1 + n] + n  a[n] == 0

In[61]:= re[15] = FindRecursion[LaguerreL[n, x]^2, n]

Out[61]= (4 - 12 n + 9 n  - 2 n  + 4 x - 4 n x + n  x)

>      2      3
a[-3 + n] + (-6 + 22 n - 21 n  + 6 n  - 14 x +

>      2      2      2      3
26 n x - 11 n  x - 7 x  + 6 n x  - x )

>      2      3
a[-2 + n] + (2 - 10 n + 15 n  - 6 n  + 6 x -

>      2      2      2      3
18 n x + 11 n  x + 5 x  - 6 n x  + x )

>      2      3      2
a[-1 + n] + (-3 n  + 2 n  - n  x) a[n] == 0

```

References

- [1] Abramowitz, M. and I. A. Stegun. 1964. *Handbook of Mathematical Functions*. Dover Publications, New York.
- [2] Hansen, E. R. 1975. *A Table of Series and Products*. Prentice-Hall, Englewood Cliffs, NJ.
- [3] Koepf, W. 1992. Power series in Computer Algebra. *J. Symb. Comp.* 13:581–603.
- [4] Koepf, W. 1993a. Examples for the algorithmic calculation of formal Puiseux, Laurent and power series. *SIGSAM Bulletin* 27:20–32.
- [5] Koepf, W. 1993b. Algorithmic development of power series. In *Artificial intelligence and symbolic mathematical computing*, ed. J. Calmet and J. A. Campbell. International Conference AISMC-1, Karlsruhe, Germany, August 1992, Proceedings, 195–213. Lecture Notes in Computer Science 737. Springer-Verlag, Berlin.
- [6] Koepf, W. Symbolic computation of formal power series with *Macsyma*. Proceedings of the *Second Workshop on Functional-Analytic Methods in Complex Analysis and Applications to Partial Differential Equations*, Trieste, Italy, January 1993. World Scientific Publishing Co., to appear.
- [7] Koepf, W. Algorithmic work with orthogonal polynomials and special functions. In preparation.
- [8] Koepf, W., and D. Schmiersau. 1994. Bounded nonvanishing functions and Bateman functions. *Complex Variables*, to appear. Preprint A/31-93, Fachbereich Mathematik der Freien Universität Berlin.

- [9] Koepf, W., and D. Schmersau. Spaces of functions satisfying simple differential equations. In preparation.
- [10] Stanley, R. P. 1980. Differentiably finite power series. *Europ. J. Combinatorics* 1:175–188.

Electronic Supplement

The electronic supplement contains the package `PowerSeries.m`.